module deque_mod
    use,intrinsic :: iso_fortran_env
    implicit none
    type deque
        integer(int64),pointer:: v(:)
        integer(int64):: l,r
        integer(int64):: lmax, rmax
    end type
    private
    public:: deque
    public:: init_deque
    public:: append, appendleft
    public:: pop, popleft
    public:: right, left
    public:: remaining_elements
    public:: remain

contains
    subroutine init_deque(dq)
        ! 初期化
        type(deque):: dq
        dq%r=0; dq%rmax=1
        dq%l=1; dq%lmax=-1
        allocate(dq%v(dq%lmax:dq%rmax))
    end subroutine


    subroutine append(dq,num)
        ! 右端に挿入
        type(deque):: dq
        integer(int64):: num
        if (dq%r+1 > dq%rmax) call add_(dq)
        dq%r=dq%r+1
        dq%v(dq%r) = num
    end subroutine

    subroutine appendleft(dq,num)
        ! 左端に挿入
        type(deque):: dq
        integer(int64):: num
        if (dq%l-1 < dq%lmax) call add_(dq)
        dq%l=dq%l-1
        dq%v(dq%l) = num
    end subroutine

    subroutine add_(dq)
        ! arrayの延長
        type(deque):: dq
        integer(int64):: l
        integer(int64),pointer:: tmp(:)
        l = size(dq%v)
        allocate(tmp(l))
        tmp(:) = dq%v(:)
        deallocate(dq%v)
        allocate(dq%v(2*dq%lmax:2*dq%rmax))
        dq%v(dq%lmax:dq%rmax) = tmp(:)
        dq%lmax = 2*dq%lmax
        dq%rmax = 2*dq%rmax
    end subroutine


    function pop(dq) result(ret)
        ! 右端から取り出し
        type(deque):: dq
        integer(int64):: ret
        ret = dq%v(dq%r)
        dq%r=dq%r-1
    end function

    function popleft(dq) result(ret)
        ! 左端から取り出し
        type(deque):: dq
        integer(int64):: ret
        ret = dq%v(dq%l)
        dq%l=dq%l+1
    end function


    function right(dq) result(ret)
        ! 右端を確認
        type(deque):: dq
        integer(int64):: ret
        ret = dq%v(dq%r)
    end function

    function left(dq) result(ret)
        ! 左端を確認
        type(deque):: dq
        integer(int64):: ret
        ret = dq%v(dq%l)
    end function

    function remaining_elements(dq) result(ret)
        ! 残りの要素数
        type(deque):: dq
        integer(int64):: ret 
        ret = dq%r - dq%l + 1
    end function

    function remain(dq) result(ret)
        ! 要素が残っているかどうか
        type(deque):: dq
        logical:: ret
        ret = remaining_elements(dq) > 0
    end function
end module


program Product_and_GCD
    use,intrinsic :: iso_fortran_env
    use deque_mod
    implicit none
    integer(int64):: n,p,cnt,i,ans, ival, inval
    type(deque):: val, nval 

    call init_deque(val)
    call init_deque(nval)

    read*, n,p

    do i=2,ceiling(sqrt(dble(p)))
        if (mod(p,i) /= 0) cycle
        call append(val,i)
        cnt=0
        do while (mod(p,i) == 0)
            p=p/i
            cnt=cnt+1
        end do
        call append(nval,cnt)
    end do

    if (p > 1) then
        call append(val,p)
        call append(nval,1_8)
    end if
    ! print*, val%v(:)
    ! print*, nval%v(:)
    ans = 1

    do while(remain(val))
        ival = pop(val)
        inval = pop(nval)
        ans=ans*ival**(inval/n)
    end do

    print'(i0)', ans
end program Product_and_GCD

Measured memory_bytes: 512